

*Watson's  
Notes*

# SPECTRUM+

**CREATIVE GRAPHICS**

D. KEDEM & I. KALISKY

**SPECTRUM+**

---

# *Watson's Notes*

**SPECTRUM+**

**I.KALISKY & D.KEDEM**

**UNIT 4**

**Creative Graphics**

**GLENTOP**  
PUBLISHERS ■ LIMITED

---

FEBRUARY 1986

All programs in this book have been written expressly to illustrate specific teaching points. They are not warranted as being suitable for any particular application. Every care has been taken in the writing and presentation of this book but no responsibility is assumed by the author or publishers for any errors or omissions contained herein.

**COPYRIGHT © Glentop Publishers Ltd 1986**

Translated from the original.

**COPYRIGHT © I.Kalisky & D.Kedem 1984, 1986**

No part of this publication may be copied, transmitted or stored in a retrieval system or reproduced in any way including but not limited to photography, photocopy, magnetic or other recording means, without prior permission from the publishers, with the exception of material entered and executed on a computer system for the reader's own use.

**ISBN 0 907792 67 7**

Published by:

Glentop Publishers Ltd  
Standfast House  
Bath Place  
High Street  
Barnet  
Herts EN5 5XE

Tel: 01-441-4130

Cover illustration from an original painting by Nick Woods



---

# Contents

	Page
Foreword	5
Chapter 1 - Introducing High Resolution Graphics	7
Chapter 2 - Video Pencil	27
Chapter 3 - Drawing Lines	35
Chapter 4 - Sines and Cosines	45
Chapter 5 - Regular Polygons	61
Summary Index	69
Appendix A - Angles and Triangles	71
Appendix B - The Sines and Cosines of Angles Greater than $90^\circ$	77
Answers	83



---

# Foreword

This Unit is called 'Creative Graphics' and is dedicated to developing 'High Resolution' graphics programs.

During the course of this unit, we will be using some trigonometry such as Sine, Cosine and PI. Don't worry if you don't know what they are as they will be explained to you when required.

Using the trigonometry and the commands taught during this unit, you will be able to design complex shapes that will in turn help you with your game writing.

Danny Kedem & Itzhak Kalisky.



# Chapter 1

## INTRODUCING HIGH RESOLUTION GRAPHICS

### Displaying Pixels

 Instruct the computer to execute the following line:

```
10 PLOT 0,0
```

The PLOT command can be found here:




 Did something appear in the bottom left-hand corner of the screen when you executed this line?

### Explanation

The PLOT command is used to display dots on the screen. Each dot is referred to as a **PIXEL** (Picture Element).

**Pixel**

 In order to see the position of the pixel on the screen, colour the border yellow.




## The Origin



Where is the pixel displayed in relation to the screen and borders?

You can see that the computer displayed the pixel in the bottom left-hand corner of the screen.



This position is called the ORIGIN and has the coordinates  $0,0$ .

**Origin**

**Note:** the coordinates of a pixel are the two numbers specified in the PLOT command.



Add this line to the program:

```
20 PLOT 100,0
```

and run the program.



Where did the computer display the additional pixel?



Where in your opinion will the computer display the following pixel?

```
30 PLOT 0,100
```



After you have thought about it, enter the line and run the program.



What coordinates must be specified in the PLOT command to complete the fourth corner of the square?



What coordinates would display a pixel in the centre of the square?

Check your answers on the computer.


## The Coordinates of a Pixel

In order for the computer to display a pixel on the screen, you must use the PLOT command and the two numbers that accompany it:

PLOT first number,second number

- The first number specifies the horizontal position of the pixel.
- The second number, the vertical position of the pixel.

These two numbers are the 'coordinates' of a pixel and together, they define the position of a pixel on the screen.



The coordinates  $(200,100)$  define pixel



The coordinates  $(200,0)$  define pixel \_\_\_\_\_.



The coordinates  $(50,150)$  define pixel \_\_\_\_\_.



The coordinates  $(0,150)$  define pixel \_\_\_\_\_.



Check your answers on the computer by adding the appropriate lines.

## Task

Erase the computer's memory and write a program that draws a line (a series of pixels) from the origin to the right-hand side of the screen. (Use  $X$  as your variable.)

**Answer 1**



What does  $X$  equal when the program stops running?



The coordinates of the bottom right-hand pixel are \_\_\_\_\_  $(255,0, 0,256, 256,0)$ .



The horizontal line displayed consists of \_\_\_\_\_  $(255, 256)$  pixels.

**Answer 2**

## Conclusion

- The coordinates of the bottom left-hand pixel are  $0,0$ .
- The coordinates of the bottom right-hand pixel are  $255,0$ .
- The screen's width consists of 256 pixels.



Now write a program that draws a vertical line. (Use Y as your vertical variable.)

After running the program, answer the following questions:

? The vertical line consists of \_\_\_\_\_ pixels.

? The coordinates of the top left-hand pixel are \_\_\_\_\_.


Check your answers on the computer.

## Task

Write a program that draws a horizontal line across the screen and then a vertical line.

**Answer 3**


Change the program so that it draws the following square:



**Answer 4**

## Task

Change the program so that a pixel is displayed every 5 pixels horizontally and every 3 pixels vertically.



### Task

Change the program again so that the computer draws the diagonals of the square.

**Hint:** you'll need to either increment or decrement both X and Y before displaying the pixel.

### Task

Write a program to draw the following pyramid where the base is 100 pixels long:



If you need a hint, look at Unit 2, Chapter 6.

**Answer 5**

### Task


Erase the computer's memory and write a two line program which continuously displays random pixels on the screen.

**Answer 6**

### Conclusion

In order to display a pixel on the screen, you must know the pixel's coordinates. The coordinates of a pixel are its horizontal and vertical distances from the bottom left-hand side of the screen.

In the example below:



X=\_\_\_\_\_ Y=\_\_\_\_\_



How many pixels does the screen consist of?


Answer 7

## Games Corner


Now that you know how to display pixels on the screen, we will write a simple game called 'Hit the Target'.

The rules of the game are as follows:


- The computer draws a randomly positioned circle (target) on the screen.



- The computer then asks you for the coordinates of one pixel inside the circle. You must estimate these coordinates



- On entering the coordinates, the pixel is displayed. In the example below, the pixel missed the target.





If you feel confident to do this program on your own, do so, however we expect most of you to follow the stages below:

### Stage 1

The first thing we must do in this game is to draw a circle. In order to do this, you will need to know the CIRCLE command:


#### CIRCLE



- Execute the following command:

```
CIRCLE 150,100,50
```

On pressing ENTER, the computer draws a circle.



The first two numbers in the CIRCLE command (150 and 100) represent the coordinates of the centre of the circle.

The third number represents the 'Radius' of the circle. (The radius of a circle is the number of pixels from the centre of the circle to the outside of the circle.)


- Now write a program that displays a circle with a radius of 10 pixels in a random position on the screen.

## Note

No point of the circle can be displayed 'off' the screen. In our program, the centre of the circle must be at least 10 pixels (radius) from the horizontal and vertical borders of the screen.

(See what happens when you try to display a point in the circle 'off' the screen.)

**Answer 8**

 Now put a GOTO command in the program to display lots of random circles.

## Stage 2

At this point, you need to INPUT the coordinates of a pixel inside the circle.

**Answer 9**

## Stage 3

The final stage of the program should display the pixel whose coordinates you input.

- If you find the game too difficult, change the border colour so that the origin position is visible.
- Add a PAUSE 0 command to the program so that on pressing any key, another circle is displayed at a random position.

**Answer 10**

## OVER 1

You may have noticed that there is a 'bug' (error) in the program. If the coordinates input are the same as any one of the pixels in the circle, you will not see the pixel displayed.

For example, execute the following command:

```
CIRCLE 100,100,20
```



What happens when you type in:

PLOT 120,100

As you can see, the pixel doesn't show because there is already a pixel at position 120,100.




Now type in:

PLOT OVER 1;120,100


OVER 1



On pressing ENTER, a pixel is removed from the circle.



Now remove the pixels from the following positions on the circle:



## Explanation

When the computer executes an OVER 1 command, it carries out the following checks:

- If there is a pixel at the coordinates specified, it gets rid of it. i.e. the pixel is displayed in the PAPER colour so it cannot be seen.
- If there isn't a pixel there, the computer displays one at the coordinates specified. i.e. the pixel is displayed in the INK colour so it can be seen.



? What will happen to the position previously erased when you execute the following command?

```
PLOT OVER 1;120,100
```



? What will the following program do?

```
10 CIRCLE OVER 1;100,100,50
20 GO TO 10
```

After thinking about it, run the program.

**Answer 11**

□ Change the program so that the circle changes colour each time it is displayed. i.e. the first circle should be black, the second blue and so on up to yellow. The process should then be repeated.

**Answer 12**

## Using OVER 1 as an Eraser


We have seen how OVER 1 can be used to erase pixels.

In the tasks that follow, you will see how OVER 1 can be implemented in your programs.

### Task

Write a program that moves a pixel across the screen from left to right.


**Answer 13**



**Note:** we discussed animation in Unit 2, Chapter 5.

## Task

Write a program that moves an arrow (a short horizontal line made up of 5 pixels) across the screen.



**Hint:** add a pixel to the right of the arrow and erase one from the left of it.

**Answer 14**

Change the OVER 1 in the program to OVER 0 and see what happens.

**OVER 0**

You can see that the computer does not erase the pixels when you use OVER 0. In fact, the computer is always in the OVER 0 state, except of course when you specify OVER 1.

## Note

In the same way that you can write the INK and PAPER commands on their own, so you can the OVER command.

```
10 OVER 1
```

If you specify OVER 1 on its own, all following PLOT commands will do the two checks. In contrast to that:

```
10 PLOT OVER 1;100,50
```

will only check the position specified in the command.

## A Bug


There is a bug in the program that moves the arrow across the screen. If you look at the left hand-side of the screen, you will see that an unwanted arrow is displayed.

This is easy to rectify with the following command:

## INVERSE 1

INVERSE 1 is also used to erase pixels but it works in a different way. It differs from the OVER command in that it will always display a pixel in the PAPER colour so it cannot be seen.

## INVERSE 1

 Add the INVERSE command to the program so that the arrow on the left of the screen is erased at the right time.

 What does the following program do:

```

10 LET X=50:LET Y=50
20 PLOT ABS X,ABS Y
30 LET X=X+1
40 LET Y=Y+1
50 IF Y=175 THEN LET Y=-Y
60 IF X=255 THEN LET X=-X
70 GO TO 20

```

In order to understand the program, you must know how to use the ABS command:

ABS means ABSolute value.

## ABS



 Execute the following command:

```
PRINT ABS -6
```



What value is displayed? \_\_\_\_\_ (-1, -6, 6)?

Generally speaking, when ABS is applied to a number, it makes it a positive number.



What happens when you apply ABS to a positive number?

```
PRINT ABS 10
```

You can see that the computer leaves the number as it is.

## Conclusion

ABS converts negative numbers to positive numbers. This is referred to in mathematics as the Absolute value.

You should now realise why the drawing changes direction when it gets close to the borders. (Explanation in Answer 15.)

**Answer 15**



How will changing the starting point of the drawing (X=50:Y=50) effect the shape displayed? Try it.



Now add the following line:

```
5 OVER 1
```

and run the program. Why, after a few minutes, are the lines erased?



Now delete the two 'ABS's in line 20 and run the program.

As you can see, it makes no difference. This is because the computer automatically takes the absolute value of the coordinates before displaying a pixel.




- Try it by executing the following command:

```
PLOT -10,-20
```


## Task

Write a program that displays the following concentric circles where each radius is 10 pixels larger than the previous one.



Answer 16

- Now change the program in memory so that the following is displayed:



**Hint:** whilst changing the size of the radius, move the centre of the circle to the right.

Answer 17

## Colouring Pixels

- Erase the computer's memory and type in and run the following line:

```
10 CIRCLE PAPER 0;INK 9;100,
100,50
```

- In order to understand what is happening, add the following line:

```
20 PLOT PAPER 3;INK 7;123,90
```

## Colouring Pixels


and run the program.

When the computer executed line 20, a magenta coloured square (PAPER 3) with a white pixel (INK 7) inside it was displayed.


 Now display another white pixel in the magenta coloured square.

**Answer 18**

Below is an enlarged diagram of the square:



You can see from the diagram that the square is constructed from 64 small squares (pixels).

 We have shown you the position of one of the pixels in the square (123,90). Now you display the pixels at the points marked by the numbers 1, 2 and 3.

**Note:** enter the commands as program lines just in case you need to change them.

**Answer 19**

The magenta coloured square is actually a character position rather than a pixel position. Therefore, when you display a pixel and specify another PAPER colour, the whole character position which the pixel is in changes colour.

 What are the coordinates of the magenta square?

**Answer 20**

 What are the coordinates of the character position which pixel 60,35 occupies?

**Answer 21**

□ Complete the following line so that the letter 'A' is displayed two character positions to the left of the magenta square:



```
65 PRINT PAPER 4;INK 9;AT
____,____;"A"
```

□ Using the PLOT command, but without displaying a pixel, colour the character position to the left of the letter A yellow.

**Answer 22**

? Do you now know why the circle was drawn with squares in line 10?

## A Surprising Result

? What would happen if you displayed a black pixel inside the magenta square?

□ Enter a line to do this and use the PAUSE command so that you can see the result.

On doing this, you will notice that all the pixels that were displayed in the square, turned black.

**Note:** within any square, the computer can only display pixels in one colour, the INK colour.


The last INK colour specified will be used to display the pixels in any one square.

□ Now make the letter 'A' disappear using the PLOT command.

**Hint:** the letter A is constructed from black pixels.

**Answer 23**

 See what happens to the pixels in a square when you display a character in it.

 What must you add to the PLOT command to make the pixels in the magenta coloured square disappear?

**Answer 24**

 What does the following program do?

```
40 LET K=0
50 FOR Y=4 TO 175 STEP 8
60 FOR X=4 TO 255 STEP 8
70 PLOT PAPER K;INK 9;X,Y
80 LET K=K+1:IF K=7 THEN LET K=0
90 NEXT X
100 NEXT Y
```

After you have thought about it, erase the computer's memory and type the program in.

 How can you erase the pixel that appears in each square?

**Answer 25**

## Task

Write the program again, however, this time your not allowed to use the PLOT command.

**Hint:** use the PRINT AT command.


**Answer 26**

 Now type in and run the following program:

```
5 LET N=0
10 FOR R=10 TO 40 STEP 2
20 CIRCLE INK N;125,85,R
30 LET N=N+1:IF N=7 THEN LET N=0
40 NEXT R
```

 Can you explain what happens?





 Change the program so that the computer draws seven different coloured circles where each circle is the minimum distance apart so as not to effect the colour of the adjacent circle.


**Answer 27**

## Summary

- You cannot display two pixels with different colours in the same square.
- If you change the colour of the PAPER when displaying a pixel, the whole character position (square) will change colour.
- Only one PAPER colour can appear in any one square.

 The INVERSE 1 command always changes the pixel's colour to that of the \_\_\_\_\_ colour of the square.

 If a pixel is displayed, the OVER 1 command will change its colour to the \_\_\_\_\_ colour of the square.


 Executing OVER 1 on a blank position will cause the pixel to change its colour to that of the \_\_\_\_\_ colour.



## Challenge


Write the following program. (The answer is not given.)

- A column of numbers (0 to 3) is displayed on the right-hand side of the screen. On the left-hand side of the screen, at a random height, an arrow is displayed.




- You must estimate where the arrow will hit the column of numbers when it moves horizontally across the screen.

- For example, we estimated 1.55:



- On inputting this number, the arrow moves across the screen.
- When the arrow hits the column of numbers, the computer displays your estimation and the correct value. In our example, we estimated 1.55 and the arrow hit the column at 1.34.



## Improvements

- The computer calculates the difference between the correct value and your estimation (ABS).
- The computer allows you to have 10 goes and at the end, displays the average difference.

## Chapter 2

# VIDEO PENCIL

Now you know how to display pixels, rewrite the 'Video Pencil' program described in Unit 3. This time, a pixel should be displayed everytime you press a cursor key rather than a square.

You have all the knowledge required to write this program. (If you have difficulties with it, look at Chapter 2, Unit 3.)

**Answer 28**

### Note

You should have a line like this in your program:

```
IF INKEY$=CHR$ 9 THEN LET X=X
+1
```

This can be shortened to the following:

```
LET X=X+(INKEY$=CHR$ 9)
```

When the computer executes the brackets:

```
(INKEY$=CHR$ 9)
```

it checks to see if the condition is true. i.e. the right cursor key has been pressed.

- If the condition is true, 1 is added to X.
- If the condition is false, i.e. the right cursor key was not pressed, 0 is added to X.



What value would the following expressions produce:

(2=3)

(2<3)

(2>3)

(2<>3)



Check your answers on the computer using the PRINT command. For example:

```
PRINT (2=3)
```



What do you think the following line does:

```
LET X=X+(INKEY$=CHR$ 9)-(INKE  
Y$=CHR$ 8)
```

It increments or decrements the variable X depending on whether you have pressed the right or left cursor key.



Now shorten the video pencil program using this technique.

**Answer 29**

## Conclusion

The tests that the computer makes on the brackets are known as TRUE and FALSE tests.

**True and False**




The expression is equal to \_\_\_\_\_(1, 0) if it is true and \_\_\_\_\_(1, 0) if it is false.

**Note:** these type of test can often replace the IF...THEN tests and as we have seen, shorten the program.

## Task

Change the video pencil program so that on pressing a key, you can move the pencil without drawing, allowing us to draw two shapes independent of each other. For example:





When you want to continue drawing again, press another key.

**Answer 30**


## Duplicating Shapes

You may remember that in Unit 2, Chapter 4, we showed you how to copy one shape all over the screen.

 Change the video pencil program so that when you have drawn a shape, for example:



you can copy it all over the screen on the press of a key.



You will need to define two arrays that hold the coordinates of the shape drawn so that the shape can be reconstructed.

(If you don't remember what an array is, look at Chapter 3, Unit 3.)

Try to write the program on your own, however, should you have any difficulties, look at the stages that we have set out below.

## Stage 1

Firstly, you must temporarily take out the lines that allow you to move the video pencil without drawing. The most efficient way of doing this is with the REM command:

## REM

The REM command allows the programmer to put in his or her own REMarks. For example:


```
REM THIS LINE DISPLAYS PIXELS
```



When the computer executes a line like this, it completely ignores the whole line and moves onto the next line. Therefore, we can use the REM command to take the necessary lines in the video pencil program, out of action.

For example:

```
28 REM PLOT OVER K;X,Y
```

 Now run the program and see what happens when you press the number '1'.

Nothing should happen, but on pressing the cursor keys, you will see that the video pencil still works.

**Note:** REM is also used to find bugs in programs. By taking lines out of action, you can see which line is causing the problem.



## Stage 2

After designing a shape, if you press a key, the computer clears the screen and starts to duplicate the shape.

This stage is quite complex and can be divided up into the following stages:

- Set up two large arrays to hold the coordinates of each pixel in the shape. The arrays should only be filled when you press a cursor key (PAUSE Ø).
- On pressing a key, the screen clears and the shape is duplicated. (See page 24.)

The full program is listed in Answer 31.

### Answer 31

**Note:** when you design a shape, keep it small otherwise the computer will display the following error message when it is duplicating the shape:

```
B Integer out of range
```

All this means is that the shape is too large to display the correct amount of times (9).

## Limiting the Size of Arrays

If you looked at the answer, you will notice that we have dimensioned the X and Y arrays to 500.

This is because the Spectrum will produce the following error message if an array is too large:

```
4 Out of memory
```

This means that the computer does not have enough memory to store an array of the size specified.

For example, type in:

```
DIM X(10000)
```

Therefore, when you are using arrays, don't make them too large because the computer's memory is limited.


## Improving the Program

Change the program so that you can reduce and enlarge the shape designed (instead of just duplicating it). For example, if you input '3', the shape is displayed 3 times larger. If you input 0.5, the shape is reduced by half.

Answer 32

## A Mirror Image

Change the program so that the computer draws a mirror image of what you are drawing. For example:



- When you draw a line to the right, the computer draws one to the left.
- When you draw a vertical line (up or down), so does the computer.

Also, add in a horizontal mirror where drawing one line causes the computer to draw three other lines. (Two in the horizontal mirror and one in the vertical mirror.)

Answer 33

## Improvements

- The computer displays the amount of pixels used in the drawing.

- Take out the REM commands in the lines that allowed you to move the video pencil without drawing and correct the bugs that are produced from it.

## Challenge

To end this chapter, type in the following program:

```
5 LET N=0
10 LET X=0:LET Y=0
50 LET A=200*RND+50
55 LET B=120*RND+50
100 LET X=X+(INKEY$=CHR$ 9)-(IN
KEY$=CHR$ 8)
120 LET Y=Y+(INKEY$=CHR$ 11)-(I
NKEY$=CHR$ 10)
150 PLOT PAPER N;INK 9;X,Y
250 LET N=(ABS(X-A)+ABS(Y-B))
/71
260 BEEP .02,8*(7-N)
290 IF INKEY$="P" THEN GO TO
350
300 GO TO 100
350 CLS
400 PLOT INK 9;A,B:CIRCLE A,B,5
420 PLOT OVER 1;X,Y:PLOT OVER 1
;X+1,Y:PLOT OVER 1;X+1,Y+1:PLOT
OVER 1;X,Y+1
```

This program is in fact a game that we have called 'Mine Detector'. You must try to find out how the game works.



## Chapter 3

# DRAWING LINES

### The Draw Command

Up until now, the only way to draw a line was by displaying lots of pixels. As you may have noticed, this process is slow.

The Spectrum computer has the ability to draw lines quickly and easily using one command.

☐ Clear the computer's memory and change the border colour to yellow.


☐ Now type in the following program and run it:

```
10 PLOT 100,50
20 PAUSE 50
30 DRAW 120,80
```



**DRAW**

You can see that the computer displays a pixel at the position 100,50, pauses for a while and then draws a line.






## Note

The computer will draw a line from the last pixel displayed. In this case, from 100,50.


The first number in the DRAW command (120) represents the horizontal distance and the second number (80) represents the vertical distance. (See diagram above.)

It is important to realise that these two numbers represent distances rather than coordinates.

 Complete the following program so that a line is drawn from the position 40,50 to the top right-hand corner of the screen:


```
5 BORDER 6
10 PLOT 40,50
20 PAUSE 50
30 DRAW _____,_____
```

**Answer 34**


 Add lines to the program so that a line is drawn from the same position (40,50) to the top left-hand corner of the screen.

Remember that the computer draws from the last pixel displayed.

**Answer 35**

 Now draw lines from position 40,50 to the remaining two corners of the screen.

**Answer 36**

 Which position will the computer display from if you erase the computer's memory and then execute the following command:

```
DRAW 100,100
```

## Explanation

As we have already said, the computer draws from the last pixel displayed. If no pixel is displayed, the computer draws from the Origin  $(0,0)$ .



Where do you think the line in the following program will be drawn from?



```
10 CIRCLE 100,100,60
20 DRAW 50,0
```

Think about it first and then run the program.

## Task

Write a program to do the following:

- The computer displays a pixel and then draws a line from this point to each corner of the screen.
- On pressing a cursor key, another pixel is displayed 5 pixels away in the direction indicated. The computer then draws lines to each corner of the screen from this point.




**Note:** use the video pencil to move the pixel.

**Answer 37**

## Task

Write a program that fills up the screen with vertical lines where each line is separated by 1 pixel.

**Hint:** use FOR...NEXT...STEP.



0 2 4 6 ...

**Answer 38**

Now fill the screen up with horizontal lines with 1 pixel separating each line.

**Answer 39**

At present, the program seems quite boring but if you increase the STEP values after displaying lines in both directions, and add in an OVER 1 command, you will see an interesting effect.


**Answer 40**

Play with the STEP value and include colour into the program to produce some more interesting effects.

**Task**


Write a program that draws lines from a point approximately in the middle of the screen to every third point on the top row of the screen.

0 3 6 9 ... 255



**Answer 41**

Now continue this process around the whole screen.



**Answer 42**

## Improvements

- Use colour.
- Add the OVER 1 command to the program.
- Increase the STEP value.



What does the following program do?

```
10 DRAW 10,0
20 DRAW 0,10
30 GO TO 10
```

## Task


Write a program to draw the largest possible rectangle on the screen without using the PLOT command.

**Answer 43**

Add a PAPER command to the DRAW command so that the rectangle is displayed as green squares.

## Task

Write a program that draws a black rectangle like the one below:



**Answer 44**

Now change the program so that the following is produced:

**Note:** there are several solutions.

**Answer 45**

### Concentric Squares

Do you remember the concentric squares program we developed in Unit 2, Chapter 6?

Now write the program again, however this time use pixels instead of squares.






Answer 46

### Task

Write a program to do the following:

- The computer draws a small square.
- The computer then draws a larger square over the first square in the OVER 1 mode. The process is then repeated with larger squares.




Answer 47


Before we go on to developing a Golf game, answer the following question.

Which commands display which of the following lines:



 Check your answers on the computer.

```
PLOT 125,85:DRAW 50,50
PLOT 125,85:DRAW -50,50
PLOT 125,85:DRAW 50,-50
PLOT 125,85:DRAW -50,-50
```





## Games Corner


With the knowledge acquired so far, we can develop a simple game rather like the game of golf.

The program should work as follows:

- The computer displays, in random positions, five small circles (holes), numbered from 1 to 5.
- Starting from the origin  $(0,0)$ , you must connect up each hole by drawing a line that ends 'in the hole'. (Input two numbers that are then used in the DRAW command.)



In the example below, the first attempt failed but the second attempt was successful!



The aim of the game is to enter each hole (in order) in the least amount of goes. (The computer should also display the amount of goes used.)

**Answer 48**

## The Doodling Computer

□ Write a program in which the computer continuously draws lines of random length and random direction. Each line should begin where the previous one ends. (Use OVER 1 to produce an interesting effect.)

### Note

The main problem when writing this program is how to ensure that the computer does not display a line off the screen, causing the program to stop.

Answer 49

### Conclusion

When using the DRAW command:


⤴ If the first number is negative, the computer will draw a line to the \_\_\_\_\_ (left, right) of the starting point.

⤴ If the second number is negative, the computer will draw a line \_\_\_\_\_ (above, below) the starting point.



### Challenge

As usual, we will end the chapter with a challenge that you can do on your own.


□ In front of you are three shapes. Write the programs to display each of them.



.1



.2



.3



## Chapter 4

# SINES AND COSINES

We will start this chapter off by showing you a graphics program which includes each of the following commands that will be discussed in this chapter:

SIN - Sine  
 COS - Cosine  
 PI - Pi



Type in the following program and run it.

```
10 FOR Z=0 TO 2*PI STEP PI/36
20 PLOT 120+60*COS Z,90+60*SIN Z
30 NEXT Z
```

You will find each of the new commands here on the keyboard:



What has the computer drawn on the screen?

As you can see, the computer has displayed pixels that together create a circle.


Using these three new commands, we will show you how to develop some good graphics programs.

### A Background Question

Before we continue with the chapter, answer the following questions:



? Below is a \_\_\_\_\_ (right-angled, obtuse-angled) triangle.




? \_\_\_\_\_ and \_\_\_\_\_ are perpendicular sides and side \_\_\_\_\_ is the hypotenuse.

? The perpendicular side opposite angle M is side \_\_\_\_\_.

If you are unable to answer these question, read Appendix A before continuing.

### Sine

In the following three right-angled triangles, B1, B2 and B3 are perpendicular sides opposite the 30 degree angle:



**Note:** 30° means 30 degrees.

□ Take a ruler and measure, in millimetres, the lengths of B1 and C1 in the smallest right-angled triangle. (Try to be as accurate as possible in your measurements.)

B1= \_\_\_\_\_  
C1= \_\_\_\_\_

Now instruct the computer to print the value of  $B1/C1$  where  $B1$  and  $C1$  equal the values measured.

Again, using the ruler, measure the lengths of the following sides:

$$B2 = \underline{\hspace{2cm}} \quad C2 = \underline{\hspace{2cm}}$$


$$B3 = \underline{\hspace{2cm}} \quad C3 = \underline{\hspace{2cm}}$$

and print the values of  $B2/C2$  and  $B3/C3$ .

On displaying the three results, you can see that they are all either very close to or equal to \_\_\_\_\_ ( $0.4$ ,  $0.5$ ,  $0.6$ ,  $0.7$ ).

**Answer 50**

Below is a right-angled triangle with an angle of  $30^\circ$ :



Without measuring, can you work out the relationship between the length of side  $B$  opposite the  $30^\circ$  angle and the length of the hypotenuse  $C$ ?

$$B/C = \underline{\hspace{2cm}}$$

Side  $B$  is half the size of side  $C$ . This is because the triangle has the same angle as the one in the three triangles on page 46 and as we have already discovered, sides  $B1$ ,  $B2$  and  $B3$  divided by  $C1$ ,  $C2$  and  $C3$  respectively, produce a value of  $0.5$  (half).

## Conclusion

In any right-angled triangle with a  $30^\circ$  angle, the ratio between the side opposite the angle and the hypotenuse is  $0.5$ .

**Note:** measurements with a ruler are inaccurate and may well produce values only approximately equal to 0.5.

## Sine

**The ratio between the side opposite the angle and the hypotenuse is called the Sine of the Angle.**

Therefore, we can say that the Sine of  $30^\circ$  is equal to 0.5, or:

$$\text{Sine } 30^\circ = \frac{\text{Side opposite angle}}{\text{Hypotenuse}}$$

$$= \frac{1}{2} = 0.5$$



Using the previous triangles, can you work out the Sine of a  $60^\circ$  angle? i.e. a right-angled triangle with a  $60^\circ$  angle in it.

## Hint

You should know from Appendix A that the angles in a triangle add up to \_\_\_\_\_ degrees.

**Answer 51**

## Conclusion

The sine of an angle, whatever the angle, is equal to the side opposite the angle divided by the hypotenuse.


In other words:

$$\text{Sine } M = \frac{B}{C}$$

where M is the angle in question.

## Exercise

Read the questions carefully and draw the triangle on a piece of paper as it is explained to you.



- In a right-angled triangle where the length of the hypotenuse is 100cm, what is the length of the side opposite the angle if the Sine of the angle is 0.3?
- In a right-angled triangle where the side opposite the angle is 50cm, what is the length of the hypotenuse if the Sine of the angle is 0.7?

**Answer 52**

## Degrees and Radians

We have seen that the Sine of  $30^\circ$  is equal to \_\_\_\_\_.

? What happens when you tell the computer to print the Sine of  $30^\circ$ :

**SIN**

```
PRINT SIN 30
```


As you can see, the computer displays the number -0.98803163.

? Why has the computer displayed this number if the Sine of  $30^\circ$  is 0.5?

**Radians**

It is simply because you and the computer were not speaking the same language. The computer does not measure angles in degrees, like we do, but in RADIANS.

Below are two angles; one represents a degree and one a radian:



1 Radian



1 Degree


As with distances, angles can also be measured in different units. e.g. distances can be measured in kilometres and centimetres etc.

Therefore, in order to display the value of Sine  $30^\circ$ , we must convert  $30^\circ$  into radians.

There are  $180^\circ$  in a straight line. (Look at Appendix A.)



How many radians are there in a straight line?



From the diagram above, we can see that there are between \_\_\_\_\_ (2 and 3, 3 and 4, 4 and 5) radians in a straight line.

In fact, the computer can give you a more accurate answer. Type in:

PRINT PI

PI



On pressing ENTER, the following number is displayed:

3.1415927

and this is the exact number of radians in a straight line ( $180^\circ$ ).



You may be wondering why there is a key to produce this number. The reason for this is that PI is a very useful number and is used in many aspects of mathematics.

Now that you know PI radians is the same as  $180^\circ$ , answer the following questions:

? An angle of  $90^\circ$  is equal to \_\_\_\_\_ (PI/3, PI/4, PI/2, PI/6) radians.

**Answer 53**

? An angle of  $60^\circ$  is equal to \_\_\_\_\_ (PI/3, PI/4, PI/2, PI/6) radians.

? An angle of  $45^\circ$  is equal to \_\_\_\_\_ radians.

? An angle of  $30^\circ$  is equal to \_\_\_\_\_ radians.

**Answer 54**

You should now know how to display the value of Sine  $30^\circ$ .

```
PRINT SIN (____)
```

You should have put PI/6 in the brackets.

**Note:** you have to use brackets now so that the computer doesn't print the Sine of the first number it finds (PI).

## Task

Write a program that allows you to input a number in degrees and on pressing ENTER, displays both the value in degrees and in radians.

(Call the degree variable M and the radians Z.)

**Answer 55**

Now improve the program in memory so that the results appear as follows:

M=60  
Z=PI/3

**Answer 56**

Now, again improve the program so that the Sine of the angle input is displayed.

**Answer 57**


## SIN 0

Tell the computer to print the value of Sine 0.

? Why is 0 displayed?


## Explanation

Below is a right-angled triangle:



As mentioned before, Sine  $Z = B/C$

Now imagine that the triangle is getting smaller and Z is getting closer to  $0^\circ$ .



You can see that as the angle gets smaller, so does the side opposite the angle. When the angle is  $0^\circ$ , side B is 0 in length, which is why the Sine of  $0^\circ$  equals 0.

$$\text{Sine } 0^\circ = 0/C$$


Sine 0 therefore equals 0.

## SIN(PI/2)

Tell the computer to print the value of Sine (PI/2).

Why was the value 1 displayed?

Imagine that angle Z is nearing PI/2 radians ( $90^\circ$ ).




The nearer angle Z gets to PI/2 radians, the closer the lengths of the hypotenuse and the side opposite the angle become. When Z equals PI/2, sides B and C are the same lengths, e.g.

$$\text{Sine (PI/2)} = 100/100$$

## Exercise


Using the diagram below, instruct the computer to display the length of the hypotenuse (B).



**Answer 58**

## Cosine

If you have understood Sine, it won't be too difficult for you to understand Cosine.




Cosine

The Cosine of angle Z is the adjacent side to the angle (A) divided by the hypotenuse (C). In other words:

$$\text{Cosine } Z = A/C$$

Using the right-angled triangles on page 47, work out the Cosine of the  $30^\circ$  angle. We advise you to use the largest triangle in order to get the most precise measurements.

Can you guess what the Cosine of  $60^\circ$  is?



$$\text{COS } 60^\circ = A/C = \underline{\hspace{2cm}}$$

COS


Now tell the computer to print the Cosine of  $60^\circ$  remembering to convert degrees to radians.

The COS command can be found here on the keyboard:



On pressing ENTER, the computer displays the number \_\_\_\_\_ which is the same as the Sine of  $30^\circ$ .

This is because a right-angled triangle with one angle equal to  $3\theta^\circ$  means that the other angle is equal to  $6\theta^\circ$ . (If you not sure why, look at Appendix A.)



$$\text{Sine } 3\theta^\circ = \text{Cosine } 6\theta^\circ$$



Do you know what the values of the following Cosines are:

$$\text{COS } \theta = \underline{\hspace{2cm}}$$

$$\text{COS}(\text{PI}/2) = \underline{\hspace{2cm}}$$



**Hint:** think about what happens to the hypotenuse and the side adjacent to the angle in question.



Check your results on the computer.

## Exercise

- What is the length of side A if Cosine  $Z = 0.37$
- What are the lengths of sides A and B in the following triangle:



**Answer 59**

## Drawing Angles



Draw a line 100 pixels long from position 40,20 at an angle of  $2\theta^\circ$ .



**Note:** the angle in question ( $2\theta^\circ$ ) is called the 'Angle of Incline'




**Answer 60**

## Task

Write a program to do the following:

- The computer displays a random size angle between  $0$  and  $90$  degrees.
- The base line should be permanently displayed on the screen whilst the hypotenuse changes depending on the angle of incline. You have to guess what the angle of incline is.
- On entering your guess, the correct angle of incline is displayed with the guess next to it.



**Answer 61**


- Improve the game so that the angle of incline is between  $0$  and  $360$  degrees.

**Hint:** change line 20.

It may surprise you that the computer can work out Sines and Cosines of angles greater than  $90^\circ$ . (If you want to know how, look at Appendix B.)

## Task

Write a program that displays six lines from a fixed position (100,100) where each line is 70 pixels long. The angle of incline should start at  $0^\circ$  and increase by  $20^\circ$  for each line drawn. The shape drawn should look like a fan.



Answer 62

- Now change the program so that a line is drawn every  $3^\circ$  through a whole circle ( $360^\circ$ ).
- Make the STEP  $1^\circ$  and include an OVER 1 command in the program to produce an interesting effect.

## Combining the Video Pencil and the Fan

- Write a program that allows you to move the position from which the lines are drawn with the cursor keys. (Use the video pencil from page 27.)

We advise you to shorten the lines to about 20 pixels in length.

Answer 63

## Games Corner

- Write a program to do the following using the program in memory:


- The computer displays two parallel lines (chimney) on the screen where the distance between them is slightly smaller than the fan (chimney sweep).



For example, the chimney is 40 pixels wide and the chimney sweep is 30 pixels wide.

- You must move the chimney sweep through the chimney without it touching the sides of the chimney.

In the example below, the player has lost because the chimney sweep went over the chimney:



**Note:** use the OVER 1 command so you can see when this happens.

## Improvements

- Make the chimney narrower.
- The chimney sweep rotates faster. (Increase the STEP value.)
- Use a winding chimney instead of a straight one.

## The Circle Program



You should now understand how the circle program at the start of this chapter works. If you don't, we suggest that you read the chapter again. If you do, continue reading.

□ Change the program on page 57 so that a pixel is displayed every  $4^\circ$  in a circle whose centre is at  $120,90$  and whose radius is  $80$  pixels.



## Challenge

□ Write the programs that display the following shapes using SIN, COS and PI:



**Note:** the answer is not given.

Handwritten scribbles and lines.

Handwritten text.

Handwritten text.




Handwritten text.






## Chapter 5

# REGULAR POLYGONS

Would it surprise you to know that the diagram below:



was constructed from one of the following REGULAR POLYGONS:



Can you guess which one of them it was?



What is a regular polygon?

**Regular Polygons**

Polygon is the name given to any shape that has sides and angles. A regular polygon has all angles equal and all sides the same length.

## Polygons

Lets see what polygons you know.

In the diagram above:

? Polygon \_\_\_\_\_ is called a TRIANGLE.

? Polygon \_\_\_\_\_ is called a SQUARE.

? Polygon \_\_\_\_\_ is called a PENTAGON.


? Polygon \_\_\_\_\_ is called a HEXAGON.

In this chapter, we will show you how to construct different shapes using regular polygons.

## Task

Write a program that draws lines of equal length where each line displayed is at a greater angle of incline than the previous one.

For example, the first angle of incline is  $0^\circ$ , the second is  $M^\circ$ , the third is  $2*M^\circ$  etc.



## Note

Before the computer draws the polygon, it asks you:

- How many lines are to be drawn. (Call the variable N.)

- The length of the lines. (Call this variable L.)
- The size of angle M in degrees. (Remember, degrees must be converted to radians.)

When the computer has received the input, it draws the lines on the screen.

**Hint:** look at the 'Fan' program on page 58. The main difference is that in the fan program, when a line is drawn, the computer returns to the centre of the fan before drawing the next line.

**Answer 64**

## Investigating the Program

Even if you don't know how the program works, on inputting the required data, a shape is produced. However, we want to display regular polygons.

## Creating Squares

Try to create a square using this program. We advise you to draw a square on a piece of paper and then decide what the values of N and M should be.

**Answer 65**

## Creating Triangles

See if you can create a triangle. Again, work on paper before inputting the data.

**Hint:** a regular triangle is called an equilateral triangle. (If you don't know what one is, look at Appendix A.)

**Answer 66**

## A Formula for Creating Polygons

- To create a triangle, N must equal 3 and M, 120.
- To create a square, N must equal 4 and M, 90.



Can you see any connection between the values of N and M in both of the above examples?

If you can't, look at the following question:



Which of the following expressions is true for both the triangle and the square:

Degrees	Radians
$M=360/N$	$Z=2*PI/N$
$M=180/N$	$Z=PI/N$
$M=N/360$	$Z=N/2*PI$

**Answer 67**

### Task

Now that you know the formula for creating polygons, draw a hexagon (6 sided shape).

**Answer 68**

### Task

If there is always a connection between two values, in this case N and M, we need not input both values.

Change the program so that the computer only requests data for the number of sides (N) and the length of the sides (L).

**Answer 69**

On making this change, see what different polygons you can produce.


## A Challenge



Is it possible to create a circle using this program? Try it.

## Task

Change the program in memory so that it displays shapes in the following way:




## A Hexagonal Snail

**Note:** you should also be able to display the snail as a pentagon or a square etc.

Answer 70

## Task

Change the program so that the following is displayed.



**Note:** no input is required at this point.



Do all the polygons have the same length sides?

Answer 71

## Rotating Polygons


By changing the program in memory, we can 'rotate' (move about a point) a polygon.

The program should work as follows:







- The computer draws a square, for example, on the screen.



- The computer then draws another square on top of the first one however, this time, the base line has an angle of incline ( $C$ ).



- The computer then continues to draw squares through  $360^\circ$ , i.e. the angle of incline should equal  $360^\circ$  when the program stops.






**Answer 72**

## Changing the Program

Make the following changes to the program in order to produce some more shapes:



- Alter the rotating angle ( $C$ ).
- Shorten the length of the sides whilst the shape rotates.

See if you can produce the shapes below:



## Task

Change the program so that the rotating point also moves in a circle as the polygons are drawn producing the following effect:



Answer 73

## A Surprise Result




Add the following line to the program:

```
1000 DRAW 10*COS(C+Z),10*SIN(C+Z)
```

and run the program.

## Challenge

See if you can write the program that displays the following shape:



We have called this the 'Shutter' drawing because it resembles the shutter in a camera.

This is quite a difficult shape to draw so in the following booklet, Unit 5, we have designated a whole chapter to its construction.

... ..

... ..



... ..

... ..

... ..

... ..

... ..



... ..

... ..

... ..

---

## SUMMARY INDEX

Below is a list of the commands and techniques taught during this unit.

Go over each one of them and check if you remember them.  
(Explanations are given on the pages stated in the brackets.)

Pixel	(7)	REM	(30)
Origin	(8)	DRAW	(35)
CIRCLE	(14)	Sine	(48)
OVER 1	(16)	SIN	(49)
OVER Ø	(18)	Radians	(49)
INVERSE 1	(19)	PI	(50)
ABS	(19)	Cosine	(54)
Colouring Pixels	(21)	COS	(54)
True and False	(28)	Regular Polygons	(61)

## Conclusion

You have now completed Unit 4 in the Spectrum series. The next booklet, called 'Journey into Memory', goes into more detail about how the computer works and shows you how to design your own characters etc. (More details can be found on the back cover of this unit.)

COMPARISON TABLE

This table compares the results of the two methods of calculation for the same data.

The first column shows the results of the first method, and the second column shows the results of the second method. The third column shows the difference between the two methods.

Year	Method 1	Method 2	Difference
1980	1000	1000	0
1981	1050	1050	0
1982	1100	1100	0
1983	1150	1150	0
1984	1200	1200	0
1985	1250	1250	0
1986	1300	1300	0
1987	1350	1350	0
1988	1400	1400	0
1989	1450	1450	0
1990	1500	1500	0

The results of the two methods are identical, as expected. This is because the two methods are based on the same data and the same calculations.




## Appendix A

# ANGLES AND TRIANGLES

In order to understand Sine and Cosine, you will need to know a little bit about trigonometry.


### Angles



? How many degrees are there in a right-angle? \_\_\_\_\_ (45, 90, 180)?

? Which of the following would be classified as an obtuse-angle? \_\_\_\_\_ (45°, 90°, 180°)?

### Note



PI radians is the same as two right-angles.

? What value (1 or 0) would be displayed for each of the following expressions:


**Note:** if you don't remember what an expression is, look at page 35.

- PI radians > right-angle
- PI radians < obtuse-angle
- Obtuse-angle > right-angle
- Right-angle > acute angled

**Answer 74**

## Vertex Angles

Below are two lines which cross each other. One of the angles is equal to  $25^\circ$ .



? What do you think the other angle (M) equals?

Two angles such as these are called vertex angles and they are equal. Therefore, M also equals  $25^\circ$ .


? Can you see two other vertex angles in this diagram?

## Parallel Angles

Below are two parallel lines, L and K.

**Note:** parallel lines are at least two lines that stay the same distance apart, like railway lines.

A third line (I) crosses the two parallel lines.



There are two angles shown in this diagram. The top angle equals  $50^\circ$ .






What do you think angle M equals?

Two angles such as these are called parallel angles and they are also equal. Therefore, M equals  $50^\circ$ .




Look at the diagram below:



Using the knowledge acquired so far, work out the value of angle M in each of the above diagrams.

## Transformed Angles

In the previous exercise, there were two angles such as the following:




The angles, which are called transformed angles, are in fact equal. (Parallel and Vertex angles.)

 Can you find any more transformed angles in this diagram?


## Triangles

In any triangle, the sum of the angles is  $180^\circ$ .  
Why?

Below is a triangle:




The triangle has three angles; M, N and P.



If you draw a line (B) parallel to side A that hits the top angle in the triangle, the sum of the angles at the top equals a straight line ( $180^\circ$ ):


$$S + P + T = 180^\circ$$

 Now, because sides A and B are parallel, angle S is the same as angle \_\_\_\_\_ (M, N) and angle T is the same as angle \_\_\_\_\_ (M, N).

Therefore, the angles in a triangle equal  $180^\circ$ .

## Equilateral Triangles

Below is an Equilateral triangle; a triangle with all sides and all angles equal:




Each angle in the triangle is equal to  $60^\circ$   
 ( $3 \times 60 = 180$ ).

? Therefore, what does angle M equal?  
 \_\_\_\_\_ ( $60^\circ$ ,  $120^\circ$ )?

## Isosceles Triangles

An Isosceles triangle is a triangle with two sides (B and C) equal:




? If the angle that these two lines create equals  $50^\circ$ , what are the values of the other two angles? \_\_\_\_\_ ( $70^\circ$ ,  $65^\circ$ ,  $130^\circ$ )?

**Note:** the two angles in question are always equal in an isosceles triangle.

## Right-Angled Triangles

A right-angle triangle is a triangle in which one of the angles equals  $90^\circ$ :



? What is the value of angle M in this right-angled triangle?

**Answer 75**

## Note


Two of the sides in a right-angled triangle are perpendicular to each other. In this triangle, sides A and B are perpendicular.

The third side, which is always opposite the right-angle, is called the hypotenuse.



? Side \_\_\_\_\_ (A, B, C) is the hypotenuse in this triangle.

? How many right-angled triangles can you find in the following diagram:



? Do all the triangles have the same angles i.e. is angle Z equal to angle M?


The answer to this question should be yes.

## Appendix B

# The Sine and Cosine of Angles Greater than $90^\circ$

### Sine

Look at the following diagram in order to understand how sine values are determined for angles greater than  $\pi/2$  radians ( $90^\circ$ ):




The Radius 'R' (hypotenuse) is at an angle of  $Z^\circ$  to the horizontal diameter.


If we drop a line from the point where the radius (R) touches the circle, we have created a right-angled triangle.

Complete the following equation to give the sine of angle Z:

$$\text{SIN } Z = \frac{?}{?}$$

**Answer 76**


 What does angle Z equal in the following diagram. (Use a protractor.)



**Note:** an angle is always measured in an anticlockwise direction. In this case, the angle is greater than  $90^\circ$ .


SIN Z = \_\_\_\_\_


**Answer 77**

 Instruct the computer to print the value of sine  $120^\circ$ . (Remember to convert degrees to radians.)


 The computer displays the value \_\_\_\_\_.

The result of B/R should be similar to the value displayed.

 What is the sine of an angle that is greater than  $180^\circ$ . For example:



Again, the sine of the angle is B/R.

 What do you think the sine of  $225^\circ$  is?

On executing `PRINT SIN(PI/4*5)`, the computer displays a negative number.

This is because side B is below the horizontal diameter of the circle. In other words, a negative value is displayed for an angle greater than  $180^\circ$ .


? Which of the following angles has a negative sine value:

$120^\circ$ ,  $260^\circ$

$300^\circ$ ,  $150^\circ$

? What happens with angles greater than  $360^\circ$ ?

As you can see in the following diagram, the triangle returns to the same position as an acute-angled triangle.



? The sine of which acute-angle is equal to the sine of  $405^\circ$ ?

The answer is  $45^\circ$  ( $405-360$ ).


? Which of the following angles will produce a negative sine value:

$460^\circ$ ,  $570^\circ$

$660^\circ$ ,  $750^\circ$

## Cosine

First of all, let's look at acute-angles:




The cosine of angle Z is the side adjacent to the angle divided by the hypotenuse.

$$\cos Z = A/R$$



How do you work out the cosine of an angle that is greater than  $90^\circ$  such as the following:




Firstly, you have to drop a line at the point where the hypotenuse (R) touches the circle. You then work out the cosine in the normal way:

$$\cos Z = A/R$$

As with sine values, cosine values can also be negative, however, a negative value is only displayed when side A is on the left of the circle.

Negative  
Cosines






## Note

Negative sine and cosine values can be useful if you are using the DRAW command. For example, when the computer executes:

```
PLOT 100,90:DRAW 70*COS(250*PI/180),70*SIN(250*PI/180)
```

which of the following lines is displayed:



Check your answer on the computer.



---

# Answers

## Answer 1

```
10 BORDER 6
20 LET X=0
30 PLOT X,0
40 LET X=X+1
50 GO TO 30
```

## Answer 2

The width of the screen is 256 pixels (0 to 255). Therefore, when X reached 256, the computer displayed the following error message:

B Integer out of range

meaning the computer could not display a pixel whose width coordinate is 256.

## Answer 3

```
10 BORDER 6
20 FOR X=0 TO 255
30 PLOT X,0
40 NEXT X
50 FOR Y=0 TO 175
60 PLOT 0,Y
70 NEXT Y
```

## Answer 4

```
10 BORDER 6
20 FOR X=80 TO 180
30 PLOT X,50
40 NEXT X
50 FOR Y=50 TO 150
60 PLOT 180,Y
70 NEXT Y
80 FOR X=180 TO 80 STEP -1
90 PLOT X,150
100 NEXT X
110 FOR Y=150 TO 50 STEP -1
120 PLOT 80,Y
130 NEXT Y
```

## Answer 5

```
10 LET A=10:LET B=110:LET Y=10
20 FOR X=A TO B
30 PLOT X,Y
40 NEXT X
50 LET A=A+1:LET B=B-1:LET Y=Y
+1
60 IF A>B THEN STOP
70 GO TO 20
```

## Answer 6

```
10 PLOT RND*255,RND*175
20 GO TO 10
```

## Answer 7

The screen consists of:

$256*176=45056$  pixels

## Answer 8

```
10 CIRCLE 235*RND+10,155*RND+10,10
```

## Answer 9

Add the following line:

```
20 INPUT "X=";X,"Y=";Y
```

## Answer 10

Add these lines:

```
40 PLOT X,Y
50 PAUSE 0
60 RUN
```

**Note:** running a program clears the screen.

## Answer 11

When you first run the program, the screen is clear so a circle is displayed. When the computer tries to display a circle for a second time, it sees that a circle is already displayed and instead, erases the circle.

## Answer 12

Note the STEP 0.5. This is used because we only want to colour every other circle, e.g. the first colour will be black (0), the second colour (0.5) doesn't matter because the circle is erased, the third colour will be blue (1) etc.

```
5 FOR K=0 TO 6 STEP 0.5
10 CIRCLE INK K;OVER 1;100,100
,50
20 NEXT K
30 GO TO 5
```



### Answer 13

```
10 FOR X=0 TO 255
20 PLOT X,85
30 PLOT OVER 1;X-1,85
40 NEXT X
```

### Answer 14

```
10 FOR X=0 TO 250
20 PLOT X+5,85
30 PLOT OVER 1;X-1,85
40 NEXT X
```

### Answer 15

We will explain the left and right movements off the screen. (The up and down movements will be similar.)

When the computer tries to display a pixel off the left-hand side of the screen, X becomes negative (0, -1, -2 etc). However, because we have used ABS in the PLOT command, X is positive (0, 1, 2 etc).

When the computer tries to display a pixel off the right-hand side of the screen, X has a value greater than 255 (256). When this happens, line 60 sets X to -256. X is then incremented:

$$-256+1=-255$$

### Answer 16

```
10 FOR R=20 TO 60 STEP 10
20 CIRCLE 125,85,R
30 NEXT R
```

## Answer 17

Add these lines:

```
5 LET X=100
10 FOR R=20 TO 60 STEP 6
20 CIRCLE X,85,R
25 LET X=X+6
30 NEXT R
```

## Answer 18

```
30 PLOT INK 7;125,93
```

## Answer 19

```
40 PLOT INK 7;120,88
50 PLOT INK 7;127,88
60 PLOT INK 7;127,95
```

## Answer 20

The purple square is positioned at 10,15.

## Answer 21

```
17,7
```

## Answer 22

```
67 PLOT PAPER 6;INK 6;100,90
```

## Answer 23

```
80 PLOT INK 4;105,90
```

## Answer 24

```
90 PLOT INK 3;123,90
```

## Answer 25

Change line 70 to:

```
70 PLOT PAPER K;INK K;X,Y
```

## Answer 26

```
40 LET K=0
50 FOR Y=0 TO 21
60 FOR X=0 TO 31
70 PRINT PAPER K;AT Y,X;" "
80 LET K=K+1:IF K=7 THEN LET K
=0
90 NEXT X
100 NEXT Y
```

## Answer 27

Change line 10 to:

```
10 FOR R=10 TO 70 STEP 10
```

## Answer 28

```
10 LET X=125:LET Y=85
20 PLOT X,Y
30 IF INKEY$=CHR$ 9 THEN LET X=
X+1
40 IF INKEY$=CHR$ 8 THEN LET X=
X-1
50 IF INKEY$=CHR$ 10 THEN LET Y
=Y-1
60 IF INKEY$=CHR$ 11 THEN LET Y
=Y+1
100 GO TO 20
```

## Answer 29

Add these lines:

```

30 LET X=X+(INKEY$=CHR$ 9)-(INKE
EY$=CHR$ 8)
40 LET Y=Y+(INKEY$=CHR$ 11)-(IN
KEY$=CHR$ 10)
100 GO TO 20

```

and delete lines 50 and 60.

## Answer 30

```

5 LET K=0
24 IF INKEY$="1" THEN LET K=1
26 IF INKEY$="0" THEN LET K=0
28 PLOT OVER K;X,Y

```

## Answer 31

```

5 LET K=0
6 LET I=1
7 DIM X(500):DIM Y(500)
10 LET X=125:LET Y=85
20 PLOT X,Y
22 IF INKEY$="P" THEN GO TO 150
24 REM IF INKEY$="1" THEN LET
K=1
26 REM IF INKEY$="0" THEN LET
K=0
28 REM PLOT OVER K;X,Y
29 LET X(I)=X:LET Y(I)=Y:LET I=
I+1:PAUSE 0
30 LET X=X+(INKEY$=CHR$ 9)-(INK
EY$=CHR$ 8)
40 LET Y=Y+(INKEY$=CHR$ 11)-(IN
KEY$=CHR$ 10)
100 GO TO 20
150 CLS:FOR X=50 TO 150 STEP 50
160 FOR Y=50 TO 150 STEP 50
200 FOR J=1 TO I-1
210 PLOT X(J)-125+X,Y(J)-85+Y
250 NEXT J
260 NEXT Y:NEXT X

```

## Answer 32

Add these lines:

```

23 IF INKEY$="C" THEN GO TO 300
300 CLS:INPUT "P=";P:FOR J=1 TO
  I-1
310 PLOT 125+(X(J)-125)*P,85+(Y
  (J)-85)*P
350 NEXT J
360 PAUSE 0:GO TO 300

```

**Note:** if you enlarge the shape, it will be displayed as dotted lines. This is because the distance between each pixel is increased depending on how many times larger you want the shape.

## Answer 33

Add this line:

```

21 PLOT 255-X,175-Y:PLOT 255-X,
  Y:PLOT X,175-Y

```

## Answer 34

```

30 DRAW 215,125

```

This is because:

```

175-50=125
255-40=215

```

## Answer 35

Add these lines:

```

40 PLOT 40,50
50 DRAW -40,125

```



## Answer 36

Add these lines:

```
60 PLOT 40,50
70 DRAW -40,-50
80 PLOT 40,50
90 DRAW 215,-50
```

## Answer 37

```
10 LET X=125:LET Y=85
40 PLOT X,Y:DRAW 255-X,175-Y
50 PLOT X,Y:DRAW -X,175-Y
60 PLOT X,Y:DRAW -X,-Y
70 PLOT X,Y:DRAW 255-X,-Y
80 LET X=X+5*(INKEY$=CHR$ 9)-5*
(INKEY$=CHR$ 8)
90 LET Y=Y+5*(INKEY$=CHR$ 11)-5
*(INKEY$=CHR$ 10)
100 GO TO 40
```

## Answer 38

```
10 FOR X=0 TO 255 STEP 2
20 PLOT X,0:DRAW 0,175
30 NEXT X
```

## Answer 39

Add these lines:

```
40 FOR Y=0 TO 175 STEP 2
50 PLOT 0,Y:DRAW 255,0
60 NEXT Y
```

**Answer 40**

Add these lines:

```
5 LET K=2:OVER 1
10 FOR X=0 TO 255 STEP K
40 FOR Y=0 TO 175 STEP K
70 LET K=K+1:GO TO 10
```

**Answer 41**

```
10 FOR X=0 TO 255 STEP 3
20 PLOT 125,85
30 DRAW X-125,90
40 NEXT X
```

**Answer 42**

Add these lines:

```
50 FOR Y=175 TO 0 STEP -3
60 PLOT 125,85:DRAW 130,Y-85
70 NEXT Y
80 FOR X=255 TO 0 STEP -3
90 PLOT 125,85:DRAW X-125,-85
100 NEXT X
110 FOR Y=0 TO 175 STEP 3
120 PLOT 125,85:DRAW -125,Y-85
130 NEXT Y
```

**Answer 43**

```
10 DRAW 0,175
20 DRAW 255,0
30 DRAW 0,-175
40 DRAW -255,0
```

**Answer 44**

```

10 FOR X=50 TO 150
20 PLOT X,40:DRAW 0,50
30 NEXT X

```

**Answer 45**

Add this line:

```

25 PLOT X,50:DRAW OVER (X>60 AND
X<140);0,30

```

**Answer 46**

```

10 LET A=10:LET B=10
30 FOR X=80 TO 0 STEP -10
40 PLOT X,X
50 DRAW 0,B:DRAW A,0:DRAW 0,-B:
DRAW -A,0
60 LET A=A+20:LET B=B+20
100 NEXT X

```

**Answer 47**

```

10 OVER 1
20 LET A=120:LET B=130:LET Y=80
40 FOR N=1 TO 10
50 FOR X=A TO B
60 PLOT X,Y:DRAW 0,B-A
70 NEXT X
80 LET A=A-5:LET B=B+5:LET Y=Y-
5
90 NEXT N

```

**Answer 48**

```

30 LET K=1
40 FOR N=1 TO 5
50 LET A=RND*240+5:LET B=RND*16
5+5
60 CIRCLE A,B,5
70 PRINT AT 21-B/8,A/8+1;N
80 NEXT N
90 PLOT 0,0
100 INPUT "X=";X,"Y=";Y
110 DRAW X,Y
120 PRINT AT 0,0;K:LET K=K+1
130 GO TO 100

```

**Answer 49**

```

20 OVER 1
30 LET A=125:LET B=85
40 PLOT 125,85
50 LET X=INT(RND*100)-50:LET Y=
INT(RND*100)-50
70 LET A=A+X:LET B=B+Y
80 IF A>255 OR A<0 OR B>175 OR
B<0 THEN LET A=A-X:LET B=B-Y:GO
TO 50
90 DRAW X,Y
100 GO TO 50

```

**Answer 50**

If your measurements were precise, the numbers displayed should be approximately 0.5.

**Answer 51**

If in a right-angled triangle, one of the angles is  $30^\circ$ , the other angle will be  $60^\circ$ :

$$180 - (90 + 30) = 60^\circ$$

Therefore:

$$\text{Sine } 60^\circ = A/C = 0.87$$

### Answer 52

If the length of the opposite side is X:

$$0.3 = X/100$$

Therefore,  $X = 30$

If the length of the hypotenuse is X:

$$0.7 = 50/X$$

$$X = 50/0.7 = 71.4$$

### Answer 53

An angle of  $90^\circ$  is equal to  $\text{PI}/2$  radians.

### Answer 54

An angle of  $60^\circ$  is equal to  $\text{PI}/3$  radians.

An angle of  $45^\circ$  equals  $\text{PI}/4$  radians.

An angle of  $30^\circ$  equals  $\text{PI}/6$  radians.

### Answer 55

```
10 INPUT "M=";M:CLS
20 PRINT AT 5,5;"M=";M
30 PRINT AT 7,5;"Z=";M*PI/180
40 GO TO 10
```

### Answer 56

Change line 30 to:

```
30 PRINT AT 7,5;"Z=PI/";180/M
```



**Answer 57**

Add the following line:

```
35 PRINT AT 9,5;"SIN Z=";SIN(M*
PI/180)
```

**Answer 58**

```
PRINT 50*SIN(35*PI/180)
```

which equals 28.68

**Answer 59**

$$A/100 = \cos Z = 0.37$$

Therefore:

$$A = 37$$

In order to find B:

$$\sin 40 = B/30$$

$$B = 30 * \sin 40$$

$$30 * \sin(40 * \pi / 180) = 19.28$$

In order to find A:

$$\cos 40 = A/30$$

$$A = 30 * \cos 40 = 22.98$$
**Answer 60**

```
10 PLOT 40,20
20 LET Z=20*PI/180
30 DRAW 100*COS Z,100*SIN Z
```

**Answer 61**

```

10 PLOT 120,20:DRAW -30,0
20 LET M=RND*90
30 LET Z=M*PI/180
40 DRAW 100*COS Z,100*SIN Z
50 INPUT "N=";N
60 PRINT AT 3,5;"M=";M
70 PRINT AT 5,5;"N=";N
80 PAUSE 0:CLS:GO TO 10

```

**Answer 62**

```

10 FOR M=0 TO 100 STEP 20
20 PLOT 100,100
30 LET Z=M*PI/180
40 DRAW 70*COS Z,70*SIN Z
50 NEXT M

```

**Answer 63**

```

40 LET M=0:LET X=125:LET Y=85
50 LET X=X+(INKEY$=CHR$ 9)-(INKEY$=CHR$ 8)
60 LET Y=Y+(INKEY$=CHR$ 11)-(INKEY$=CHR$ 10)
70 PLOT X,Y
80 LET Z=M*PI/180
90 DRAW 20*COS Z,20*SIN Z
100 LET M=M+10
120 GO TO 50

```

**Answer 64**

```

40 INPUT "N=";N
50 INPUT "L=";L
60 INPUT "M=";M
70 LET Z=M*PI/180
80 PLOT 100,20
90 FOR I=1 TO N
100 DRAW L*COS((I-1)*Z),L*SIN((I-1)*Z)
110 NEXT I

```

**Answer 65**

Input the following to display a square:

N=4      M=90

**Answer 66**

Input the following to display a triangle:

N=3      M=120

**Answer 67**

M=360/N      Z=2\*PI/N

**Answer 68**

In order to display a hexagon, input the following values:

N=6      M=60

**Answer 69**

Change line 70:

```
70 LET Z=360/N*PI/180
```

and delete line 60.

**Answer 70**

Add these lines:

```
90 FOR I=1 TO 6*N
105 LET L=L-1
```

## Answer 71

Add these lines:

```
40 LET L=40
50 FOR N=3 TO 8
90 FOR I=1 TO N
120 NEXT N
```

and delete line 105.

## Answer 72

Add these lines:

```
40 INPUT"L=";L
50 INPUT"N=";N
60 FOR C=0 TO 2*PI STEP PI/6
80 PLOT 125,85
100 DRAW L*COS(C+(I-1)*Z),L*SIN
(C+(I-1)*Z)
120 NEXT C
```

## Answer 73

Add these lines:

```
60 FOR C=0 TO 2*PI STEP PI/12
80 PLOT 125+20*COS C,85+20*SIN
C
```

## Answer 74

1. 1
2. 0
3. 1
4. 1

**Answer 75**

The sum of the angles in a triangle equals  $180^\circ$ . Therefore:

$$M = 180 - (90 + 37) = 53$$


Right-angle

**Answer 76**

$$\sin Z = B/R$$

**Answer 77**

$$\sin Z = B/R$$





**This series of self-instruction books will teach you the secrets of writing programs in BASIC on your SPECTRUM+ computer.**

#### **Unit 1: FIRST STEPS IN BASIC**

Starting with the first things every programmer needs to know, you will learn to issue commands to the computer, as well as writing and running programs. By the end of the unit you'll be able to make your computer perform useful and interesting tasks.

#### **Unit 2: EXPLORING BASIC**

This unit teaches you the most important concepts of BASIC: numeric variables, string variables, FOR . . . NEXT and IF . . . THEN statements, and much more. You'll create a digital computer clock, and interesting graphics programs including animation.

#### **Unit 3: COMPUTER GAMES**

In this unit you will learn to develop various computer video games. As you progress through the unit, new programming concepts such as random numbers will be introduced. By the end you'll have considerable programming skills.

#### **Unit 4: CREATIVE GRAPHICS**

This unit will enable you to continue to expand your programming skills. You'll move step by step through essential mathematical concepts, like sine and cosine, for use in creating amazing on-screen graphics.

#### **Unit 5: JOURNEY INTO MEMORY**

This unit takes you deep into the computer's memory. You'll learn about how memory is organised, how to design your own characters to replace the standard ones, and how to produce your own graphics symbols.

#### **Unit 6: ADVANCED PROGRAMMING**

This unit is intended to develop a variety of programs using the knowledge you have accumulated in the first four units. Although bringing together the skills already gained there is still plenty to learn. The important data storage concept of two dimensional arrays is covered, allowing you to develop a large program called Word Game.

**And after that . . . we're planning more units to deal with special subjects, such as 3D graphics, machine code and more.**

**£3.95**

**GLENTOP**  
PUBLISHERS LIMITED

Glentop Publishers Ltd.,  
Standfast House, Bath Place,  
High Street, Barnet, Herts. EN5 1ED  
Tel: 01-441 4130

ISBN 0-907792-67-7



9 780907 792673